

PLOTTING MCCABE-THIELE DIAGRAMS IN MICROSOFT EXCEL

John L. Gossage
Dan F. Smith Department of Chemical Engineering
Lamar University

Abstract

For a binary distillation problem involving a constant relative volatility, a McCabe-Thiele diagram is useful for demonstrating the relationships among the reflux ratio, the number of stages, the relative volatility, and the feed condition. However, these diagrams are difficult and time-consuming to draw accurately by hand. This paper provides directions for plotting McCabe-Thiele diagrams using Microsoft Excel.

Notation

Variable	Definition
A	More volatile binary component
alpha, α	Relative volatility
D	Distillate molar flowrate
eta, η	Murphree efficiency
F	Feed molar flowrate
N	Number of stages
q	Feed "liquid fraction" (see Eq. 3)
r	Calculated value (see Eq.'s 7 and 8)
R	Reflux molar flowrate
RR	Reflux ratio = D/R
x	Liquid mole fraction of A
y	Vapor mole fraction of A
z	Feed mole fraction of A
Subscripts	
b	Bottoms
d	Distillate
eq	Equilibrium
f	Feed
p	Point where feed line intersects equilibrium curve
min	Minimum

Introduction

McCabe-Thiele diagrams are helpful for teaching distillation to students in Mass Transfer (or Unit Operations) class. However, teachers know that drawing an accurate McCabe-Thiele diagram, especially freehand, is quite difficult and rather time-consuming. Thus, this paper details a method for generating a McCabe-Thiele diagram using Microsoft Excel.

For a good overview of the McCabe-Thiele method as applied to binary distillation problems, see Geankoplis [1], pages 706-724.

For a good introduction into the intricacies of Microsoft Excel 2007 (which is the version used in this paper) see *Excel 2007, The Missing Manual*, by MacDonald [2].

As shown in Figure 1 below, the user inputs the constant relative volatility, α ; the mole fraction of A in the feed, z_f ; the feed "liquid fraction", q ; the desired mole fraction of A in the distillate and bottoms, x_d and x_b ; the reflux ratio, RR; and the Murphree efficiency, η . The Excel spreadsheet then outputs the point (x_p , y_p) where the feed line intersects with the equilibrium line, the minimum reflux ratio RR_{min}, the point (x_f , y_f) where the feed line intersects with the operating lines, the required number of equilibrium stages, and the feed stage. Excel then draws the McCabe-Thiele diagram, showing the feed line, the rectifying and stripping operating lines, the equilibrium curve, and the stages.




Figure 1: Excel Spreadsheet showing user inputs, outputs, and the McCabe-Thiele diagram.

Assumptions

The spreadsheet's calculations are based on the following four assumptions, in addition to those normally assumed for McCabe-Thiele analysis. First, the distillation is assumed to be a binary distillation with components A and B, where A is assumed to be the more volatile of the two species. Second, the relative volatility, α , is assumed to be constant throughout the column. The relative volatility is defined as follows [1] (equation 11.3-1, page 701):

$$\alpha = \frac{y/x}{(1-y)/(1-x)} \quad \text{Eq. 1}$$

The third assumption is that the Murphree efficiency is constant for the whole column. The fourth assumption is that the number of stages is no more than 50: this limit places a maximum on the size of the output table used to generate the plot.

Equations

Several equations are needed to provide enough information to be able to generate the McCabe-Thiele diagram. Figure 2 shows more

detail on the McCabe-Thiele diagram, indicating the names of the various lines, some of their slopes, and the locations of points "P" and "F".




Figure 2: The McCabe-Thiele diagram. Point P is the intersection of the Feed Line and the Equilibrium Curve; point F is the intersection of the Feed Line with the Rectifying and Stripping Operating Lines.

The equilibrium curve is generated by first rearranging Eq. 1 to solve for y as shown in [1] (equation 11.3-4, page 701):

Equil curve:
$$y = \frac{\alpha \cdot x}{1 + (\alpha - 1) \cdot x} \quad \text{Eq. 2}$$

The equation for the feed line depends on the value of q , which is defined as follows [1] (equation 11.4-12, page 710):

$$q = \frac{\left(\begin{array}{l} \text{heat needed to vaporize 1 mole} \\ \text{of feed at entering conditions} \end{array} \right)}{\left(\begin{array}{l} \text{molar latent heat of} \\ \text{vaporization of feed} \end{array} \right)} \quad \text{Eq. 3}$$

Thus, $q=0$ means the feed is a saturated vapor; $q=1$ means the feed is a saturated liquid; $0 < q < 1$ means the feed is a saturated liquid-vapor mixture; $q < 0$ means the feed is a superheated vapor, and $q > 1$ means the feed is a subcooled liquid. The feed line can then be drawn starting from the point (z_f, z_f) with a slope given by $q/(q-1)$. Therefore, if $q \neq 1$ the feed line equation is given by this equation:

Feed line:
$$y = \frac{q \cdot x - z_f}{q - 1} \quad \text{Eq. 4}$$

At the point $P(x_p, y_p)$ the equilibrium curve and the feed line intersect. Thus, setting Eq. 2 equal to Eq. 4 and solving gives the value of $x=x_p$, and then substituting this x_p back into Eq. 2 gives $y=y_p$. A bit of algebra produces the following equations for x_p , depending on q :

If $q=1$
$$x_p = z_f \quad \text{Eq. 5}$$

If $q=0$
$$x_p = \frac{z_f}{\alpha - z_f \cdot (\alpha - 1)} \quad \text{Eq. 6}$$

Other
$$x_p = \frac{r + \sqrt{r^2 + 4 \cdot z_f \cdot (\alpha - 1) \cdot q}}{2 \cdot (\alpha - 1) \cdot q} \quad \text{Eq. 7}$$

 q 's

Where r is given by this equation:

$$r = (\alpha - 1) \cdot (z_f + q) - \alpha \quad \text{Eq. 8}$$

In addition, the value of y_p also depends on q :

If $q=1$
$$y_p = \frac{\alpha \cdot z_f}{1 + (\alpha - 1) \cdot z_f} \quad \text{Eq. 9}$$

If $q=0$
$$y_p = z_f \quad \text{Eq. 10}$$

Other
$$y_p = \frac{\alpha \cdot x_p}{1 + (\alpha - 1) \cdot x_p} \quad \text{Eq. 11}$$

 q 's

Once the location of point $P(x_p, y_p)$ is known, the spreadsheet next calculates the minimum reflux ratio, RR_{min} . As shown in Figure 3, if the value of RR is decreased to RR_{min} , then points P and F merge.




Figure 3: The McCabe-Thiele diagram at minimum reflux, where P and F merge.

The spreadsheet can calculate the slope of the minimum reflux op-line, since the locations of its two endpoints are now known:

$$S_{min} = \frac{x_D - y_P}{x_D - x_P} \quad \text{Eq. 12}$$

Given this value, the spreadsheet next calculates the minimum reflux ratio as follows:

$$RR_{min} = \frac{S_{min}}{1 - S_{min}} \quad \text{Eq. 13}$$

Combining Eq. 12 with Eq. 13 and solving for RR_{min} gives the following equation:

$$RR_{min} = \frac{x_D - y_P}{y_P - x_P} \quad \text{Eq. 14}$$

The spreadsheet next calculates the point F by means of the following formulas. For x_f , the equation depends on the value of q , as shown:

If $q=1$ $x_F = z_F$ Eq. 15

If $q \neq 1$ $x_F = \frac{x_D / (1 + RR)^{-z_F / (q - 1)}}{q / (q - 1) - RR / (1 + RR)}$ Eq. 16

The value of y_f can then be found:

$$y_F = \frac{x_D + x_F \cdot RR}{1 + RR}$$
 Eq. 17

Next, the spreadsheet needs to calculate the intersection point for each of the equilibrium stages with the appropriate operating line, as shown in Figure 4:




Figure 4: The intersection points 0 to 5 between each equilibrium stage and the appropriate operating line.

As shown in Figure 4, this McCabe-Thiele diagram has 6 intersection points, numbered 0 through 5. Thus, for this problem the spreadsheet has to calculate 6 (x,y) points. Point 0 is easy: it is simply (x_D,x_D). The value of each of the other x_i's can be calculated first by rearranging Eq. 2 to solve for $x=x_{eq}$ as follows:

$$x_{eq} = \frac{y_{i-1}}{\alpha - y_{i-1} \cdot (\alpha - 1)}$$
 Eq. 18

Thus, if the Murphree efficiency is 1, then $x_i = x_{eq}$; if not, then Eq. 15 can be combined with the definition of the Murphree efficiency:

$$\eta = \frac{x_{i-1} - x_i}{x_{i-1} - x_{eq}}$$
 Eq. 19

to yield the following equation for x_i :

$$x_i = x_{i-1} - \eta(x_{i-1} - x_{eq})$$
 Eq. 20

This allows the spreadsheet to draw the horizontal line (representing the “tread” of the step) from the point (x_{i-1},y_{i-1}) to (x_i,y_{i-1}).

Once x_i is known, the spreadsheet calculates the value of y_i by drawing a vertical line (the “riser” of the step) from (x_i,y_{i-1}) down to the appropriate operating line. Since the operation of the column changes from rectifying to stripping as the value of x falls below x_f, the choice of operating line depends on the value of x_f, as shown:

$x_i > x_f$ $y_i = \frac{x_D + x_i \cdot RR}{1 + RR}$ Eq. 21

$x_i < x_f$ $y_i = \left(\frac{y_F - x_B}{x_F - x_B} \right) (x_i - x_B) + x_B$ Eq. 22

Excel

The Excel spreadsheet can now be constructed with the use of the above equations. Here are the 58 simple steps in the process.

1. **Enter the data shown below into the indicated cells.** To type the entries that span more than one row and/or column, select the appropriate cells, then click “Merge & Center” (on Home/Alignment). To enter the Greek letters and mathematical symbols, click Insert/Symbol. To draw borders around the cells, select the cells and click Home, then select the Borders icon in the Font box.

	A	B	C	D
1	Inputs		Range	
2	α	4	$\alpha > 1$	
3	zf	0.7	$x_b < z_f < x_d$	
4	q	0.4		
5	xd	0.95	$z_f < x_d < 1$	
6	xb	0.1	$0 < x_b < z_f$	
7	RR	1.3	$> RR_{min}$	
8	η	1	$0 < \eta \leq 1$	
9				
10	Outputs			
11	xp		Intersection of feed line with equil line	
12	yp			
13	RRmin			
14	xf		Intersection of feed line with op lines	
15	yf			
16	Stages		$N \leq 50$	
17	Nf		Feed Stage	

2. Name the “alpha” cell in B2. Click on cell B2, then enter the name “alpha” in the Name Box, as shown below in Figure 5:




Figure 5: The name “alpha” has been entered into the Name Box, as shown.

3. Name the other variable cells in Column B. Use the names “zf”, “q”, “xd”, “xb”, “RR”, “eta”, “xp”, “yp”, “RRmin”, “xf”, “yf”, and “N” in rows 3 through 8 and 11 through 16. The purpose of naming these cells is so that the names can be used in the subsequent equations on the spreadsheet. Any mistakes made while entering the names can be corrected by clicking on Formulas/Name Manager (or Ctrl-F3).

4. Enter the following formula in cell B11 to calculate xp by Eq.’s 5, 6, 7, and 8:

$$=IF(q=1, z_f, IF(q=0, z_f / (\alpha - z_f * (\alpha - 1)), ((\alpha - 1) * (z_f + q) - \alpha) + SQRT(((\alpha - 1) * (z_f + q) - \alpha)^2 + 4 * z_f * (\alpha - 1) * q)) / (2 * (\alpha - 1) * q))$$

5. Enter the following formula in cell B12 to calculate yp by Eq.’s 9, 10, and 11:

$$=IF(q=1, \alpha * z_f / (1 + z_f * (\alpha - 1)), IF(q=0, z_f, \alpha * x_p / (1 + x_p * (\alpha - 1))))$$

6. Enter the following formula in cell B13 to calculate RRmin by Eq. 14:

$$= (x_d - y_p) / (y_p - x_p)$$

7. Enter the following formula in cell B14 to calculate xf by Eq.’s 15 and 16:

$$=IF(q=1, z_f, (x_d / (1 + RR) + z_f / (q - 1)) / (q / (q - 1) - RR / (1 + RR)))$$

8. Enter the following formula in cell B15 to calculate yf by Eq. 17:

$$= (x_d + x_f * RR) / (1 + RR)$$

9. Enter the following formula in cell B16 to calculate N, the number of stages:

$$=IF(RR > RR_{min}, (MAX(A21:A71) - 1 + (VLOOKUP(MAX(A21:A71) - 1, (A21:B71), 2) - x_b) / ((VLOOKUP(MAX(A21:A71) - 1, (A21:B71), 2) - VLOOKUP(MAX(A21:A71), (A21:B71), 2))))), "INF")$$

Note that this (and the next) step will give a result of “#N/A” until after step 16 is done.

This formula checks to see if $RR > RR_{min}$. If so, it returns “INF” (for infinity) for the number of stages. If not, it calculates the whole number of stages plus the fraction of the final stage needed to reach an x-value equal to x_b . In the present case, Figure 1 shows N to be 4.96740. More details about this formula (and the one below in Step 10) will be given below in the “Excel Formula Explanation” section.

10. Enter the following formula in cell B17 to calculate Nf, the feed stage:

=IF(RR>RRmin, (INDEX(A21:B71, MATCH(xf, B21:B71, -1), 1)+1), "INF")

Now the calculations for each stage can be performed.

11. Type the following data into the indicated cells on the spreadsheet, including the 0:

	A	B	C
19	Stages		
20	#	x	y
21	0		
22			

12. In cells B21 and C21, enter the following:

=xd

This sets both x_0 and y_0 equal to x_D .

13. In cell A22, enter this formula:

=IF(A21<>"", (IF(C21>xb, A21+1, "")), "")

This increments the Stage Number, *unless* the preceding Stage Number (cell A21) is blank or the previous value of y_i (cell C21) is less than x_B .

14. Type the following formula in cell B22 to calculate x_i by Eq.'s 18 and 20:

=IF(A22<>"", (B21-eta*(B21-C21/(alpha-C21*(alpha-1))))), "")

Note that if the current Stage Number (cell A22) is blank, this formula leaves the value of x_i blank as well.

15. Type the following formula in cell C22 to calculate y_i by Eq.'s 21 and 22:

=IF(A22<>"", (IF(B22>xf, (xd+B22*RR)/(1+RR), ((yf-xb)/(xf-xb)*(B22-xb+xb))), ""))

Note that if the current Stage Number (cell A22) is blank, this formula leaves the value of y_i blank as well; otherwise, it calculates y_i by Eq. 21 if $x_i > x_f$, else it uses Eq. 22.

16. Now copy cells A22 to C22 all the way down to C71. Click once on cell A22, then while holding the left mouse button, drag across to select cells A22 to C22, then type Ctrl-C to copy these three cells. Now type

Ctrl-G to open the “Go To” dialog box (as shown in Figure 6), enter “c71” in the Reference box, then, *while holding the Shift key*, click the OK button. This will select the entire range from A22 to C71. Now type Ctrl-V to paste the equations. Note: Excel provides other methods of performing this “copy & paste” task, but when the final destination is known, this method is convenient to use. This Excel “secret” is described by MacDonald in [2], page 82.




Figure 6: The “Go To” dialog box.

17. Now check the Output values in Figure 1 to ensure that all the preceding formulas have been entered correctly. Also check the Stages values shown here:

	A	B	C
19	Stages		
20	#	x	y
21	0	0.95000	0.95000
22	1	0.82609	0.87996
23	2	0.64698	0.77873
24	3	0.46803	0.57379
25	4	0.25181	0.29544
26	5	0.09488	0.09341
27			
28			

Note that cell A27 is blank: that's because the y-value for the preceding Stage (Stage 5), 0.09341, is smaller than x_B , 0.1.

Now the data need to be arranged in a table so that Excel can plot the McCabe-Thiele diagram. The secret to drawing plots with multiple lines in Excel is that all the data should be in the same table (not absolutely required, but convenient), with the "x-axis" variables all in a single column. The various "y-axis" variables are then lined up vs. the appropriate x-axis variables, with spaces left for unneeded values.

18. Type the following headings in Row 19, one per column, starting in Column D and ending in Column N:

#, x , 45° line, Rect line, Strip line, Feed line, Eq line, Stages, Distillate, Feed, Bottoms

Column E (the one labeled " x ") will be the x-axis variable in the plot. Columns F through N will be the nine different lines that will be plotted to form the McCabe-Thiele diagram. Column D, the "#", is for the Stage Numbers.

19. In cell L20 (under "Distillate"), enter a 0. In cells E20, E21, and L21, enter the following formula:

=xd

These four cells represent two (x,y) points to draw the dashed line from the x-axis up to the 45° line at $x=x_D$.

20. In cell M22 (two rows under "Feed"), enter a 0. In cells E22, E23, and M23, enter the following formula:

=zf

These four cells represent two (x,y) points to draw the dashed line from the x-axis up to the 45° line at $x=z_F$.

21. In cell N24 (four rows under "Bottoms"), enter a 0. In cells E24, E25, and N25, enter the following formula:

=xb

These four cells represent two (x,y) points to draw the dashed line from the x-axis up to the 45° line at $x=x_B$.

22. Enter "0" in cells F26 (six rows under "45° line"), and E26. Enter "1" in cells E27 and F27.

These four cells represent two (x,y) points that will be connected to draw the 45° line.

23. In cells E28 and G28 (eight rows under "Rect line"), enter this formula:

=xd

24. In cell E29 enter this formula:

=xf

25. In cell G29 enter this formula:

=yf

These four cells (from steps 23-25) represent two (x,y) points that will be connected to draw the Rectifying operating line.

26. In cell E30 enter this formula:

=xf

27. In cell H30 (ten rows beneath the "Strip line" heading) enter this formula:

=yf

28. In cells E31 and H31 enter this formula:

=xb

These four cells (from steps 26-28) represent two (x,y) points that will be connected to draw the Stripping operating line.

29. In cells E32 and I32 (twelve rows below the "Feed line"), enter this formula:

=zf

30. In cell E33 enter this formula:

=xp

31. In cell I33 enter this formula:

=yp

These four cells (from steps 29-31) represent two (x,y) points that will be connected to draw the Feed line.

32. Now comes the Equilibrium curve. In cell E34 enter "0". In cell J34 (fourteen rows beneath the "Eq Line" heading) enter the following formula from Eq. 2:

= alpha*E34/(1+E34*(alpha-1))

33. In cell E35 enter the following formula:

= E34+0.025

34. Copy cell J34 into cell J35.

35. Now copy cells E35:J35 all the way down to cell J74. This gives 41 points for Excel to use to draw the Equilibrium curve.

36. Finally, the Stage data are transferred onto the table. In cell D75 enter this formula:

= A21

37. In cell E75 enter this formula:

= IF(D75<>"", VLOOKUP(D75, \$A\$21:\$C\$71, 2), 0)

38. In cell K75 (55 rows below the "Stages" heading) enter this formula:

=IF(D75<>"", IF(D75<N, VLOOKUP(D75, \$A\$21:\$C\$71, 3), E75), 0)

39. In cell E76 enter this formula:

=IF(D75<>"", VLOOKUP(D75+1, \$A\$21:\$C\$71, 2), 0)

40. In cell K76 enter this formula:

=IF(D75<>"", IF(D75<N, VLOOKUP(D75, \$A\$21:\$C\$71, 3), E75), 0)

Yes, exactly the same formula is entered into both K75 and K76. One way to save some typing is to type the formula into K75 and then hit "Enter". Then click once more on cell K75, move the cursor to the "Formula Entry" box (just to the right of the "Name Box"), highlight the entire formula, hit Ctrl-C to copy it, then hit "Esc" (to avoid inadvertently changing the formula in K75). Then click on cell K76 and hit Ctrl-V to paste the formula.

41. Leave cell D76 blank. In cell D77 enter this formula:

= IF(D75<>"", (IF(D75< MAX(\$A\$21:\$A\$71), D75+1, "")), "")

Here, the Stage # will be incremented by 1 as long as the following two criteria are true: the previous Stage # (cell D75) is *not* blank, and the previous Stage # is less than the

largest value among the Stages in Column A. If either criterion is not met, then the current Stage # will be blank. Note that if the current Stage # is blank, then the formulas given above in steps 37-40 will give values of "0".

42. Highlight cells E75:K76, hit Ctrl-C to copy, then click cell E77 and hit Ctrl-V to paste. Row 75 represents the "tread" of the step, while row 76 is its "riser".

43. Now copy cells D77:K78 all the way down to cell K176. Note that most of the rows will just contain 0's: in the present problem, only rows 86 and above will be non-zero. The remaining rows are present for problems that contain many more stages (up to 50).

Excel Formula Explanation

Several of the formulas entered above make use of advanced functions built-in to Excel: in particular, MAX, VLOOKUP, INDEX, and MATCH. The first, MAX, is pretty straightforward: given a range of cells to check, it returns the largest value. Thus, the formula "MAX(A21:A71)" finds the largest stage number in Column A: in the present problem, that's the "5" in A26.

The VLOOKUP function takes three required arguments plus an optional one. The first argument is the *Lookup_value* that Excel tries to find in the *Table_array* (the second argument). The third argument then tells Excel the *Col_index_num* (or Column Index Number, in English). The fourth (optional) argument is a logical variable: if TRUE (or omitted), then Excel finds either an exact match or the largest value that is smaller than the *Lookup_value*; if FALSE, then Excel takes only an exact match. Thus, in cell E77 a portion of the formula said VLOOKUP(D77, \$A\$21:\$C\$71, 2). Here, cell D77 contains the *Lookup_value* ("1", in the present problem). The *Table_array* is \$A\$21:\$C\$71, so Excel looks in Column A between rows 21 and 71 inclusive for the first "1" that it can find. Finally, the *Col_index_num* is 2, so the formula returns the value in the same row as the "1" that

it found, but in the second column (that's Column B) of the `Table_array`. In the present problem, the first "1" in the range given is in row 22, so the formula returns the value "0.82609", which is the value in cell B22.

One drawback of `VLOOKUP` is that the `Lookup_value` must be in the leftmost column of the `Table_array`. Thus, while it works admirably when searching for the Stage # (as used here), it does not work "backwards": it can search for the Stage # and return the corresponding x-value (or even the y-value, as used in K77), but it cannot search for the x-value and return the Stage #, because the Stage # appears to the left of the x-value in the `Table_array`. To perform such a "left lookup" requires the use of two Excel commands used together. A portion of the formula in cell B17 contains two Excel functions `INDEX` and `MATCH`:

```
INDEX(A21:B71,MATCH(xf,B21:B71,-1),1)
```

The `MATCH` function takes three arguments: the `Lookup_value`, the `Lookup_array`, and the `Match_type`. In the example here, the `Lookup_value` is `xf` (0.61176 in the present problem) and the `Lookup_array` is Column B between rows 21 and 71, inclusive. The `Match_type=-1` means that the formula is looking for the smallest value that is greater than or equal to the `Lookup_value`, and the formula assumes that the values in the `Lookup_array` are arranged in descending order. Thus the formula looks for the smallest value among the x-values in the Stages that is larger than (or equal to) `xf`. In the present problem, that x-value is 0.64698 in cell B23. Since cell B23 is item number 3 in the `Lookup_array`, the formula returns a value of 3.

The `INDEX` function takes three arguments, the `array`, the `row_num`, and the `column_num`, and returns the value found in the `array` at the specified `row_num` and the `column_num`. In the present problem, the `MATCH` function returns a value of 3 for the `row_num`, and since the `array` is specified as `A21:B71` and the `column_num` is given as 1, the formula returns the value found

in the first column of the third row of the `array` `A21:B71`, that is the value found in cell A23, 2. The formula entered into B17 then adds 1 to this result to find the feed tray, which in the present problem is 3.

This method of performing a "left lookup" is detailed by MacDonald in [2], pages 338-339.

Drawing the McCabe-Thiele Diagram

Now that the data have been placed in the appropriate tabular form, Excel can now draw the McCabe-Thiele diagram. Here are the steps in the process.

44. **Select cells E20:N176.** A simple way is to click on cell E20, then use the "Go To" shortcut described above in step 16, using a Reference of N176.
45. **Click the "Insert" tab on the Ribbon, and in the "Charts" region click the arrow below "Scatter" and select "Scatter with Straight Lines".** This will draw the chart.
46. **Move the chart so its upper-left-hand corner is in cell F1.** To move the chart, point the cursor at any edge of the chart (*not* the edge of the graph itself). The cursor will change to a four-way arrow. Click and drag the chart to the desired location.
47. **Remove the horizontal gridlines.** Click on any of the values on the y-axis: Excel will then select all of them. Right-click and select "Format Major Gridlines...". Click "Close". Now all the horizontal gridlines will be selected. Hit the "Delete" key to delete all of them.
48. **Format the y-axis.** Click on any of the values on the y-axis: Excel will then select all of them. Right-click and select "Format Axis...". On "Axis Options", after "Maximum" click the "Fixed" radio button and enter a 1. After "Major unit" click the "Fixed" radio button and enter 0.1. Below "Axis Options" click "Number". From the

Category list choose Number and set the Decimal places to 1. Click “Close”.

49. **Format the x-axis.** Click on any of the values on the x-axis: Excel will then select all of them. Right-click and select “Format Axis...”. On “Axis Options”, after “Maximum” click the “Fixed” radio button and enter a 1. After “Major unit” click the “Fixed” radio button and enter 0.1. Select “Number”. From the Category list choose Number and set the Decimal places to 1. Click “Close”.
50. **To delete the Legend, click on it and hit the delete key.**
51. **Resize the chart.** Click the resize handle (it looks like three dots in a horizontal line) at the bottom center of the chart: the cursor changes to a two-headed vertical arrow. Click and drag so the bottom edge of the chart falls just above the Row 19 headings.
52. **Make the “Distillate” line dashed rather than solid.** Right-click on the line and select “Format Data Series...”. In the “Format Data Series” dialog box, select Line style, then select Dash type and choose “Round Dot”. Click “Close”.
53. **Repeat step 52 twice more: once for the Feed line, and again for the Bottoms line.**
54. **Smooth the Equilibrium curve.** Right-click on the curve and select “Format Data Series...”. In the “Format Data Series” dialog box, select Line style, then check the “Smoothed line” box and click “Close”.
55. **Make both the 45° and Stages lines black.** Right-click on the 45° line, and select “Format Data Series...”, select Line color, then select the Solid line radio button. In the Color box, click the arrow and choose black. Repeat for the Stages line.
56. **Add the x-axis title.** Click on the chart, then on the Ribbon select “Layout” (just below

the Chart Tools). In the “Labels” area click the arrow below “Axis Titles”. Select “Primary Horizontal Axis Title”, then select “Title Below Axis”. This will create a text box with the text “Axis Title” under the x-axis. Click in the text box, replace the given text with “xA”, then highlight just the “A”, right-click, choose “Font”, and check the “Subscript” box.

57. **Repeat step 56 for the y-axis title.** Select “Rotated Title” for the y-axis.
58. **Compare the resulting plot with that shown in Figure 1 to ensure that all of the steps have been completed satisfactorily.**

Does the Condenser Count as a Stage?

As described by Geankoplis [1] on pages 723-724, if the condenser is a partial condenser (so the distillate product is removed as a vapor), and the partial condenser reaches equilibrium, then the condenser counts as an equilibrium stage. For the more-common case of a total condenser, the condenser does not perform a separation so it does not count as an equilibrium stage.

Extensions of this Method

The spreadsheet described here can be used to find the minimum number of stages, N_{min} . To do this, simply set the Reflux Ratio RR to a very high value, say 1,000,000. Then, the rectifying operating line will have a slope of 0.999999, which is close enough to one to represent total reflux. The calculated number of stages would then be equal to N_{min} . For the values of α , z_f , q , x_d , x_b , and η given in Figure 1, $N_{min} = 3.8066\dots$

The above minimum stage number result combined with the earlier minimum reflux ratio result (depicted in Figure 3) demonstrates the following important conclusions about the relationship between number of stages, reflux ratio, and the economics of the process:

N	∞	Const. : Cost	$0 < N_{\min} < N < \infty$
			$\Updownarrow \quad \Updownarrow \quad \Updownarrow$
RR	∞	Utility : Cost	$\infty > RR > RR_{\min} > 0$
			$\Updownarrow \quad \Updownarrow \quad \Updownarrow$
Econ.		Total : Cost	$\infty > \$ < \infty$

The more stages, the more expensive the column will be to construct; the higher the RR, the higher the utility costs (to cool the condenser and heat the reboiler) will be. Thus, somewhere between minimum stages and minimum reflux will be the economically optimum conditions for the column (shown qualitatively by “\$”). This insight helps the students in Plant Design.

If the Excel Solver add-in is used, then the spreadsheet can also be used to calculate the required RR to obtain a specified number of stages for given values of the other inputs. For the values of α , z_f , q , x_d , x_b , and η given in Figure 1, to have a total of 6 stages, the RR must be 0.80324....

Increasing Maximum Number of Stages

The model as described has a “50-stage” limit. This can easily be increased, say to 60 stages, by changing all of the range references of A21:A71 to A21:A81, A21:B71 to A21:B81, and \$A\$21:\$B\$71 to \$A\$21:\$B\$81 (thus, increasing the number of rows by 10), copying cells A71:C71 down to C81, and copying cells D176:N176 down to N196 (thus increasing the length of the plot table by 20 rows). To update the chart, click on the graph: Excel will then highlight the plot table. Grab the lower-right-hand corner box and drag it down to cell N196.

References

1. Geankoplis, Christie John, *Transport Processes and Separation Process Principles*, 4th Edition, Prentice-Hall Professional Technical Reference, Upper Saddle River, NJ, 2003.
2. MacDonald, Matthew, *Excel 2007: The Missing Manual*, Pogue Press; Revised edition, 2007.

Biographical Information

John L. Gossage is an Associate Professor in the Dan F. Smith Department of Chemical Engineering at Lamar University. His main research areas are simulation, applied probability, and engineering education. He currently teaches simulation and kinetics classes at both the undergraduate and graduate levels. He holds a Ph.D. in chemical engineering from Illinois Institute of Technology in Chicago, IL.

Note: Requests for a copy of the Excel spreadsheet may be emailed to the author at che_prof@hotmail.com